AD-A055 324 AIR FORCE AVIONICS LAB WRIGHT-PATTERSON AFB OHIO F/G 9/5

TRANSIENT RESPONSE OF FILTERS.(U)

DEC 77 J B TSUI, J E ADAIR, J E HAWKINS

UNCLASSIFIED AFAL-TR-77-249 NL

1 OF 1

AD
A055 324

END

DATE FILMED

8-78

DDC



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS 1963-A

FOR FURTHER TRAN

AD A055324

AFAL-TR-77-249

TRANSIENT RESPONSE OF FILTERS

Passive ECM Branch
Electronic Warfare Division

December 1977
TECHNICAL REPORT AFAL-TR-77-249
Interim Report for the Period October 1975 to November 1977

Approved for public release; distribution unlimited

AIR FORCE AVIONICS LABORATORY
AIR FORCE WRIGHT AERONAUTICAL LABORATORIES
Air Force Systems Command
Wright-Patterson Air Force Base, Ohio 45433

NOTICE

When Government drawings, specifications, or other data are used for any purpose other than in connection with a definitely related Government procurement operation, the United States Government thereby incurs no responsibility nor any obligation whatsoever; and the fact that the government may have formulated, furnished, or in any way supplied the said drawings, specifications, or other data, is not to be regarded by implication or otherwise as in any manner licensing the holder or any other person or corporation, or conveying any rights or permission to manufacture, use, or sell any patented invention that may in any way be related thereto.

This report has been reviewed by the Information Office (OI) and is releasable to the National Technical Information Service (NTIS). At NTIS, it will be available to the general public, including foreign nations.

This technical report has been reviewed and is approved for publication.

James B. Y. Tsui
J. B. Y. TSUI, PhD
Project Engineer

William F. Bahret
WILLIAM F. BAHRET
Chief, Passive ECM Branch
Electronic Warfare Division

FOR THE COMMANDER

Joseph H. Jacobs
JOSEPH H. JACOBS, Colonel, USAF
Chief, Electronic Warfare Division

"If your address has changed, if you wish to be removed from our mailing list, or if the addressee is no longer employed by your organization please notify AFAL/WRP, W-PAFB, OH 45433 to help us maintain a current mailing list".

Copies of this report should not be returned unless return is required by security considerations, contractual obligations, or notice on a specific document.

AIR FORCE/56780/1 May 1978 — 50

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER AFAL-TR-77-249	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) TRANSIENT RESPONSE OF FILTERS		5. TYPE OF REPORT & PERIOD COVERED Interim Rep[redacted]t, [redacted] Oct 75 [redacted] Nov 77
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) J.B.Y. Tsui, [redacted] J.E. Adair, [redacted] [redacted] J.E. Hawkins, [redacted] S.J. LaFleur		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS Air Force Avionics Laboratory (WRP) Wright-Patterson AFB, OH 45433		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS 62204F 7633 76331115
11. CONTROLLING OFFICE NAME AND ADDRESS Air Force Avionics Laboratory (WRP) Wright-Patterson AFB, OH 45433		12. REPORT DATE Dec[redacted]77
		13. NUMBER OF PAGES 44
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)		15. SECURITY CLASS. (of this report) Unclassified
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Filter Response
Transient Response
Computer Modeling

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

One of the most important factors in designing receivers is to determine the transient response of the signal after it passes filters. It is time consuming and costly to measure the transient response experimentally. This report presents the theoretical analysis and a computer program to calculate filter transient response. Experimental data from both conventional and surface acoustic wave filters have also been presented to verify the validity of the program. The program can handle both Butterworth and Chebyshev filters and their combinations in cascade. The input signals can be a square, a composite, and a sine square wave. The computer program that calculates the transient response is also listed.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

011 670

FOREWORD

This report presents the theoretical analysis and a computer program to predict the transient response of band-pass filters with different input signals. This is an interim report in the continuing effort to evaluate channelized receivers.

This technical report was prepared by Dr. J.B.Y. Tsui of the Passive Electronic Countermeasures Branch, Electronic Warfare Division, and Dr. J.E. Adair of Microwave Technology Branch, Electronic Technology Division, The Air Force Avionics Laboratory, Wright-Patterson AFB, Ohio, under Project 7633, Task 1115. Mr. J.E. Hawkins of Systems Research Laboratory and Mrs. S.J. LaFleur of the Digital Programming Branch, ASD Computer Center, wrote the computer program for this report.

The authors wish to express their appreciation to K.R. Laker of the Air Force Cambridge Research Laboratory for the valuable discussions with him, and to C.P. Poirier of the ASD Computer Center and C. Gulley of Systems Research Laboratory for their help in the programming.

The work period for this effort extended from October 1975 to November 1977.

TABLE OF CONTENTS

PRECEDING PAGE BLANK

TABLE OF ILLUSTRATIONS

TABLE OF ILLUSTRATIONS (CONTD)

SECTION I

INTRODUCTION

One of the most powerful methods of measuring frequency of several simultaneous signals, with limited frequency resolution required, is by using a bank of contiguous filters. If the incoming signal is continuous wave (CW), the frequency is relatively easy to determine, since the insertion loss characteristics of the filter are known. The relative outputs of two or more adjacent filters then identify frequency. If the incoming signal is pulse modulated, the output is a complex function of time, and requires more sophisticated detection schemes. To optimize the filter bank and the detection scheme, one has to know the output from different types of filters. It is time consuming and expensive to build the circuits and measure their performance (Reference 1).

The purpose of this report is to present an analytical procedure with computer calculations which determines the time response of the output signals, given the filter transfer characteristics. With this method, a large number of filters can be analyzed for the design of a channelized receiver without going through the construction and test phase. To document the validity of the analytical calculations, they are compared with measured performance of various filters.

The computer program is written to handle several kinds of band-pass filters, including Butterworth, Chebyshev, and a combination of two of these in cascade. The three kinds of pulsed carrier input signals used in the calculations are a square, a composite, and a sine square. The square pulse is assumed to have infinitely sharp leading and trailing edges. The composite pulse has finite leading and trailing edges that can be selected independently. The sine square pulse is commonly used to approximate a short signal with slow rising and trailing edges.

SECTION II
MATHEMATICAL MODEL

In the calculations, the transfer functions of the filters are first predicted from the insertion loss versus frequency measurements; then the time domain impulse response of the transfer function is derived. The computer program then calculates the convolution of the input signals and the filter transfer function.

1. FILTERS

The filters are either specified or measured and their transfer function analytically fitted to the data, using a Butterworth or a Chebyshev model. The basic equation used to determine the number of poles is

$$V^2 = \frac{1}{1 + \left(\frac{\omega^2-\omega_0{}^2}{\omega\Delta\omega}\right)^{2n}} \qquad (1)$$

where V is the output voltage of the transfer filter function, ω is the angular velocity, and ω_o is the center angular velocity of the filter. For a Butterworth filter, the poles are located as follows:

$$P_k = \Delta\omega \; [\cos \; (\frac{2k-1+n}{2n} \; \pi) + j \; \sin \; (\frac{2k-1+n}{2n} \; \pi)] \qquad (2)$$

where

$k = 1,2,\ldots,2n$,

$\Delta\omega = 2\pi \times$ bandwidth of the filter,

n = number of poles of the filters.

Only P_k with positive real parts are kept, and the poles corresponding to a band-pass filter are

$$S_k = \frac{P_k}{2} \pm j\omega \tag{3}$$

For the Chebyshev filters,

$$P_k = \Delta\omega\ (\tanh v \sin u_k + j\cos u_k) \tag{4}$$

$$u_k = \frac{\pi}{2n}\ (2k-1) \tag{5}$$

$$v = \frac{1}{n} \sinh^{-1}\left(\frac{1}{\varepsilon}\right) \tag{6}$$

and

$$\varepsilon = \sqrt{10^{R_f/10} - 1} \tag{7}$$

where $k = 1,2,\ldots,2n$,

n = number of poles of the filters,

R_f = ripple factor in dB,

ω = 2π × signal frequency, and

ω_o = 2π × filter center frequency.

Similarly, only P_k with positive real part are kept and the locations of the poles are calculated by equation (3). The transfer function of the filters is given as

$$H(s) = \frac{(\Delta\omega)^n S^n}{(S+S_1)(S+S_2)\ldots(S+S_{2n})}$$

$$= \frac{A_1}{S+S_1} + \frac{A_2}{S+S_2} + \ldots + \frac{A_{2n}}{S+S_{2n}} \tag{8}$$

and the corresponding impulse response is

$$H(t) = A_1 e^{-S_1 t} + A_2 e^{-S_2 t} + \ldots + A_{2n} e^{-S_{2n} t} \tag{9}$$

2. SIGNALS

The general shape of the input signal can be written as

$$R(t) = R_0(t) \cos \omega t = \frac{R_0(t)}{2} \left[e^{j\omega t} + e^{-j\omega t} \right] \tag{10}$$

where $\omega = 2\pi \times$ the carrier frequency of the signal and $R_0(t)$ = envelope of the input signal. For a square wave envelope input from $t = 0$ to $t = t_1$,

$$R_0(t) = U(t) - U(t-t_1) \tag{11}$$

where $U(t)$ is the step function. For a composite signal as shown in Figure 1,

$$R_0(t) = DtU(t) - D(t-t_1)U(t-t_1) - E(t-t_2)U(t-t_2) + E(t-t_3)U(t-t_3) \tag{12}$$

where $D = \frac{1}{t_1}$ and $E = \frac{1}{t_3 - t_2}$.

For a sine square input,

$$R_0(t) = \sin^2 \frac{\pi t}{t_1} U(t) - \sin^2 \frac{\pi t}{t_1} U(t-t_1) \tag{13}$$

SECTION III

OUTPUT

The output from the filter is given by the convolution integral

$$Y(t) = \int_{o}^{t} R(\tau)H(t-\tau)d\tau \tag{14}$$

1. SQUARE INPUT SIGNAL

For the square input, Equation 14 becomes

$$Y(t) = U(t)\int_{o}^{t} R(\tau)H(t-\tau)d\tau - U(t-t_1)\int_{t_1}^{t} R(\tau)H(t-\tau)d\tau \tag{15}$$

For $t < t_1$

$$Y(t) = \sum_{k=1}^{2n}\left[Y_k(t) - Y_k(0)\right] \tag{16}$$

and for $t \geq t_1$

$$Y(t) = \sum_{k=1}^{2n}\left[Y_k(t_1) - Y_k(0)\right] \tag{17}$$

where

$$Y_k(\tau) = \frac{A_k e^{-S_k t}}{2}\left[\frac{e^{B_k \tau}}{B_k} + \frac{e^{C_k \tau}}{C_k}\right] \tag{18}$$

$$B_k = S_k + j\omega$$

and

$$C_k = S_k - j\omega$$

2. COMPOSITE SIGNAL

For the composite signal, the output for $t < t_1$ becomes

$$Y_{01}(t) = \sum_{k=1}^{2n} \Big[DX_k(t) - DX_k(0) \Big] \tag{19}$$

where

$$X_k(\tau) = \frac{A_k}{2} e^{-S_k t} \left[e^{B_k \tau}\left(\frac{\tau}{B_k} - \frac{1}{B_k{}^2}\right) + e^{C_k \tau}\left(\frac{\tau}{C_k} - \frac{1}{C_k{}^2}\right)\right] \tag{20}$$

for $t_1 \le t < t_2$

$$Y(t) = \sum_{k=1}^{2n} \Big[-DX_k(0) + DX_k(t_1) + Dt_1 Y_k(t) - Dt_1 Y_k(t_1) \Big] \tag{21}$$

for $t_2 \le t < t_3$

$$Y(t) = \sum_{k=1}^{2n} \Big[-DX_k(0) + DX_k(t_1) + Dt_1 Y_k(t) - Dt_1 Y_k(t_1) - EX_k(t) + EX_k(t_2) + Et_2 Y_k(t) - Et_2 Y_k(t_2) \Big] \tag{22}$$

and for $t_3 \le t$

$$Y(t) = \sum_{k=1}^{2n} \Big[-DX_k(0) + DX_k(t_1) + Dt_1 Y_k(t) - Dt_1 Y_k(t_1) + EX_k(t_2) + Et_2 Y_k(t) - Et_2 Y_k(t_2) - EX_k(t_3) - Et_3 Y_k(t) + Et_3 Y_k(t_3) \Big] \tag{23}$$

3. SINE SQUARE WAVE SIGNAL

For the sine square wave, the output for $t < t_1$ becomes

$$Y(t) = \sum_{k=1}^{2n} \left[Y_k(t) - Y_k(0) \right] \tag{24}$$

and for $t \leq t_1$

$$Y(t) = \sum_{k=1}^{2n} \left[Z_k(t_1) - Z_k(0) \right] \tag{25}$$

where

$$Z_k(\tau) = \frac{A_k e^{-S_k t}}{2} \left\{ \frac{e^{B_k \tau}}{\left(\frac{B_k t_1}{\pi}\right)^2 + 4} \left[\frac{B_k t_1}{\pi} \sin^2\left(\frac{\pi\tau}{t_1}\right) - 2\sin\left(\frac{\pi\tau}{t_1}\right) \cos\left(\frac{\pi\tau}{t_1}\right) + \frac{2\pi}{B_k t_1} \right] + \frac{e^{C_k \tau}}{\left(\frac{C_k t_1}{\pi}\right)^2 + 4} \left[\frac{C_k t_1}{\pi} \sin^2\left(\frac{\pi\tau}{t_1}\right) - 2\sin\left(\frac{\pi\tau}{t_1}\right) \cos\left(\frac{\pi\tau}{t_1}\right) + \frac{2\pi}{C_k t_1} \right] \right\} \tag{26}$$

A computer program which calculates Equations 16 through 26 and plots the results has been written and is shown in the appendix. The following section of this report discusses the computer program.

SECTION IV

COMPUTER PROGRAM

The flow diagram of the computer program shown in Figure 2 is presently set up to handle a maximum of $n = 12$ poles, but with minor changes it can handle a larger number of poles. Since only the envelope of the output is of interest, the program plots the envelope rather than the RF signal inside the pulse. The program will pick the minimum of $2\pi/\omega$ and $2\pi/\omega_0$ and use it as a standard time value. The first thirty local maximums are found using steps of $1/10\pi$ times the standard value. In order to minimize computation time, subsequent points are found by incrementing 0.8 of the standard value from the last maximum and then using steps of $1/10\pi$ times the standard value to find the next maximum. The estimated error in amplitude is

$$1 - \cos\left(\frac{1}{2} \cdot \frac{360}{10\pi}\right) = 0.005$$

or 0.5% with respect to the true maximum value.

SECTION V

CALCULATED AND EXPERIMENTAL RESULTS FOR CONVENTIONAL FILTERS

In order to cover all the possibilities, seven cases are calculated with the results as shown in Figure 3a through Figure 5. In these Figures, only the envelopes of the signals are plotted and their amplitudes are normalized according to the peak value of each individual output. In Figures 3a, 3b, and 3c, a 3-pole Chebyshev filter is used with a center frequency of 300.8 MHz, bandwidth of 17.8 MHz and a ripple factor of 0.25 dB. The input signals are all the composite waveform with $T_1 = 15$ ns, $T_2 = 190$ ns and $T_3 = 200$ ns. The center frequencies of the input used are

301 MHz, 309.8 MHz, and 317 MHz, respectively. In Figures 4a, 4b, and 4c, a 5-pole Butterworth filter is used having a center frequency of 300.5 MHz and bandwidth of 15.8 MHz. The input signals are all square wave with a pulse width T of 300 ns. The center frequencies of the input signals are 300.5 MHz, 291.4 MHz, and 292.6 MHz, respectively. In Figure 5, the same 5-pole Butterworth filter is used. The input signal is a sine square with T_1 = 20 ns and center frequency of 300.5 MHz. The amplitudes of all the outputs are normalized to unity and the times used are either 50 ns or 100 ns as shown in the figures. For a verification of the calculated results, experimental measurements are made for the same input waveforms. The RF pulse was generated by a RF switch and the output displayed on a Tectronix type 7904 oscilloscope.

Figures 6 and 7 show the insertion loss vs frequency of the three filters used for the analytical calculations above. The input signals are shown in Figures 8, 9, and 10. Figure 8 shows the composite signal, with T_1 = 15 ns, T_2 = 190 ns, and T_3 = 200 ns; Figure 9 shows the square wave signal of T = 300 ns; Figure 10 is the sine square signal of 20 ns. The outputs are shown in Figures 11a through 13. The frequencies are the same as specified in Figures 3a through 5. Figures 11a through 12a match the calculated results very well. However, the results of Figures 12b and 12c do not match Figures 4b and 4c as closely. By changing the frequency from 291.4 to 292.86 MHz and from 292.6 to 293.444 MHz as shown in Figures 12d and 12e, they match the calculated results better. Obviously, the frequency measurement, or the bandwidth of the filter measured, is off slightly. At the center of the filter the shape of the output signal is not very frequency dependent, but at the edge of the filter the output depends very strongly on the frequency. In Figure 13, the measured result agrees with the calculated results rather well. However, the input signal, as shown in Figure 10, has some reflections in the time domain; and is not a true sine square signal. Thus, the output may deviate slightly from the ideal case.

SECTION VI

APPROXIMATION OF SURFACE ACOUSTIC WAVE (SAW) FILTER

A SAW filter with an insertion loss vs frequency as shown in Fig. 14 is used as the filter. This SAW filter was built by a cosine square time domain configuration on a pedestal (Ref. 3). This filter cannot be properly approximated by a single Butterworth or Chebyshev filter. However, it can be represented by two conventional filters in cascade. Both filters used in the analysis are Butterworth with the same center frequency of 343.5 MHz. One is a 7-pole filter with bandwidth of 9.5 MHz, the other is a single pole with bandwidth of 7.0 MHz. The input signal is a composite one with T_1 = 25 ns, T_2 = 320 ns, and T_3 = 340 ns. The center frequencies are 343.5 MHz, 335.6 MHz, and 353 MHz. The calculated and measured results are shown in Figures 15a, b, c, and 16a, b, c, respectively. In Figure 15a, c, and 16a, c, the results match fairly well; however, in Figure 15b and 16b the relative amplitudes between the two peaks are reversed in the calculated and measured results. It is suspected that the approximation is not close enough and may require a better combination of cascaded filters. For some SAW filters, the analytical model using Butterworth or Chebyshev filters must be modified by including the phase response in order to obtain the true time domain performance. A number of SAW filters are presently being measured to completely characterize their frequency domain performance and subsequently their time domain response.

SECTION VII

CONCLUSION

For the limited number of cases illustrated above, the computer program does generate a very accurate result for use in predicting the output of a filter. This program can be used to design receivers with filters, especially a channelized receiver which uses many filters. This program can also be extended to handle low-pass, high-pass, and band-rejection filters by using the appropriate filter transfer characteristics (Reference 2).

It also demonstrated that the transient response of some SAW filters can be approximated by this approach. The particular SAW filters that may be used in this approach are those displaying linear phase response in addition to providing Butterworth or Chebyshev amplitude response in the frequency domain.

REFERENCES

1. Hollis, R., "Double Detection Filter Techniques Test Report," Watkins-Johnson Co., Palo Alto, CA, 1973.

2. Chausi, M.S., "Principles and Design of Linear Active Circuits," Chapters 4, 15, McGraw-Hill, Inc., 1965.

3. Szabo, T.L., and Slobodnik, A.J., Jr., "Diffraction Compensation and Periodic Apodized Acoustic Surface Wave Filters," IEEE Trans. on Sonics and Ultrasonics U. SU-21, April 1974, pp 114-119.



Fig. 1 Envelope of the composite input signal



Fig. 2 Computer program functional flow diagram



Fig. 2 Computer program functional flow diagram (Con't)



Fig. 2 Computer program functional flow diagram (Con't)

Fig. 3 Calculated output of a 3-pole Chebyshev filter with f_0 = 300.8 MHz f = 17.8 MHz, RF = 0.25 dB, the input signal is composite with T_1 = 15 ns, T_2 = 190 ns and T_3 = 200 ns



3 a. Input frequency 301.1 MHz



3 b. Input frequency 309.8 MHz



3 c. Input frequency 317.0 MHz

Fig. 4 Calculated output of a 5-pole Butterworth filter with f_0 = 300.5 MHz

f = 15.8 MHz and square input signal of T = 300 ns



4 a. Input frequency 300.5 MHz



4 b. Input frequency 291.4 MHz



4 c. Input frequency 292.6 MHz



Fig. 5 Output of the 5-pole Butterworth filter with a sine squared input of T = 20 ns and f = 300.5 MHz



f_0 = 300 MHz 20 MHz/div 10 dB/div

Fig. 6 Insertion loss vs frequency of a 3-pole Chebyshev filter



f_0 = 300 MHz 20 MHz/div 10 dB/div

Fig. 7 Insertion loss vs frequency of a 5-pole Butterworth filter



Fig. 8 Composite input signal with T_1 = 15 ns, T_2 = 190 ns and T_3 = 200 ns



Fig. 9 Square signal T = 300 ns



Fig. 10 Sine squared signal T = 20 ns

Fig. 11 Measured output of the 3-pole Chebyshev filter



11 a. Input frequency 301 MHz



11 b. Input frequency 309.8 MHz



11 c. Input frequency 310 MHz

Fig. 12 Measured output of the 5-pole Butterworth filter



12 a. Input frequency 300.5 MHz



12 b. Input frequency 291.4 MHz



12 c. Input Frequency 292.6 MHz



12 d. Input frequency 292.86 MHz



12 e. Input frequency 293.44 MHz



Fig. 13 Measured output of the 5-pole Butterworth filter with a sine squared input signal



f_0 = 343.5 MHz 20 MHz/div 10 dB/div

Fig. 14 Insertion loss vs frequency of the SAW filter

Fig. 15 Calculated output from the SAW filter



15 a. Input frequency 343.5 MHz



15 b. Input frequency 335.6 MHz



15 c. Input frequency 353.0 MHz

Fig. 16 Measured output from the SAW filter



16 a. Input frequency 343.5 MHz



16 b. Input frequency 335.6 MHz



16 c. Input frequency 353.0 MHz

APPENDIX

COMPUTER PROGRAM LISTING

```
      PROGRAM FILTER (INPUT,OUTPUT,TAPE4,TAPE5=INPUT,TAPE6=OUTPUT)
C-----------------------------------------------------------------------
C  THIS PROGRAM PROCESSESS AN INPUT SIGNAL THROUGH AN N POLE FILTER.
C    THE OUTPUT SIGNAL HAS A RELATIVE AMPLITUDE AS A FUNCTION OF TIME
C    THE PROGRAM IS WRITTEN IN FORTRAN IV EXTENDED FOR A CDC6600.
C    PLOTTING IS DONE ON AN OFFLINE CALCOMP PLOTTER.
C-----------------------------------------------------------------------
      COMPLEX X1,X2,X3,X4,X5,X6,X7,X8,X88
      COMPLEX D4
      COMPLEX X61,X62,X63,X71,X72,X73
      COMPLEX A(48),B(48),C(48),OMEGAI,OMEGAJ,P(12),S(48),ST,ST2,TERM4,
     1Y(3),SS(24)
      DOUBLE DELT,T
      LOGICAL IY,IE,IFA,IEND
      LOGICAL IFP
      DIMENSION AH(2000),DATA(1024),TH(2000)
      DIMENSION FF(5),SF(5)
      COMMON/VARS/AH,AMPT,DELT,IVAR,L,M,T,TH,TMAX,XVAR,XVAR1,
     1YVAR,Y,DELT2,IFP,IFA
      NAMELIST/VALUE/KIS,SCF,T1,T2,T3,KF,NP,FCF,BW,RF,TMAX,CT,IVAR,XVAR,
     1 YVAR,IY,IE,IFA,IEND,ICF,IFP
      NAMELIST/FIL2/KF,BW,NP,RF,FCF
      DATA IVAR,IEND,IY,IE,IFA,TMAX,PI,XVAR,YVAR/0,.T.,3*.F.,0.,
     1 3.14159265358979,7.75,2.5/
      SINHI(X)=ALOG(X+SQRT(X**2+1.))
      ICF=1
C-----------------------------------------------------------------------
C    EACH FILTER SIMULATED MUST BE SPECIFIED BY A /VALUE/ NAMELIST.
C    NAMELIST /FIL2/ CAN ONLY BE SPECIFIED FOR A COMPOSITE FILTER AND
C    MUST IMMEDIATELY FOLLOW ITS RELATED FILTER'S /VALUE/ INPUT.
C   THE INPUT PARAMETERS ARE INPUTED THROUGH NAMELIST/VALUE/ AND ARE AS
C     FOLLOWS:
C       KIS   = "KIND OF INPUT SIGNAL", MAY HAVE A VALUE FROM 1 THROUGH 3
C               IF 1, THEN THE SIGNAL IS A SQUARE WAVE
C               IF 2, THEN A RAMP TO PEAK, PEAK FOR SOME TIME, AND THEN A
C                       NEGATIVE SLOPE TO 0 AMPLITUDE AT SOME LATER TIME
C               IF 3, THEN THE WAVE IS APPROXIMATED BY SINE SQUARED
C       SCF   = "SIGNAL CENTER FREQUENCY" IN HERTZ
C       T1    = TIME DURATION OF SIGNAL FOR KIS = 1 OR 3 AND FOR KIS = 2
C                 IT IS THE END OF THE RAMP (PEAK)
C       T2    = TIME DURATION OF PEAK SIGNAL + T1 FOR KIS = 2 ONLY
C       T3    = TIME DURATION OF NEGATIVE SLOPE + T1 + T2 FOR KIS = 2
C                 ONLY
C       KF    = "KIND OF FILTER", MAY HAVE A VALUE OF 1 OR 2
C               IF 1, THEN THE FILTER IS A BUTTERWORTH
C               IF 2, THEN THE FILTER IS A CHEBYSHEV
C       NP    = NUMBER OF POLES FOR THE FILTER (PRESENTLY LIMITED TO 12)
C       FCF   = "FILTER CENTER FREQUENCY" IN HERTZ
C       BW    = BANDWIDTH OF THE FILTER (BETWEEN -3 DB POINTS)
C       RF    = RIPPLE FACTOR FOR CHEBYSHEV FILTER
C       TMAX  = MAXIMUM TIME DURATION OF OUTPUT SIGNAL (SEE NOTE BELOW)
```

```
C       CT     = MULTIPLIER WHICH IS THE NUMBER OF TIMES THE OUTPUT
C                SIGNAL EXCEEDS THE INPUT SIGNAL (SEE NOTE BELOW)
C     IVAR  = NUMBER OF ITERATIONS THAT ARE USED IN CALCULATING Y(I)
C     XVAR  = LENGTH OF X-AXIS OF PLOT (INCHES) INITIALLY; THIS VALUE
C                IS RECALCULATED IF THE AH(I) OR AL(I) ARRAYS ARE TOO
C                LARGE
C     YVAR  = HEIGHT OF Y-AXIS OF PLOT (INCHES); DOES NOT CHANGE
C       ICF=1    SINGLE FILTER
C       ICF=2    COMPOSITE FILTER
C      IY     = DECISION WHETHER TO WRITE OUT Y ARRAY
C               IF .TRUE. THEN THE ARRAY IS WRITTEN OUT
C               IF .FALSE. THEN THE ARRAY IS NOT WRITTEN OUT
C      IF     = DECISION WHETHER TO WRITE OUT ENVELOPS OF Y ARRAY
C               IF .TRUE. THEN THE ARRAY IS WRITTEN OUT
C               IF .FALSE. THEN THE ARRAY IS NOT WRITTEN OUT
C      IFP      =FREQUENCY PLOT ---PLOT IS DONE IF .TRUE.
C      IFA   = DECISION WHETHER TO WRITE OUT FREQUENCY ARRAY
C              IF .T. THEN ARRAY IS PRINTED
C              IF .F. THEN ARRAY IS NOT PRINTED
C      IEND  = DECISION WHETHER THERE IS ANOTHER SET OF DATA TO BE RUN
C               IF .TRUE. THEN ANOTHER SET IS RUN
C               IF .FALSE. THEN THE PROGRAM TERMINATES
C       NOTE:    IF TMAX IS SET AT 0, THEN TMAX = AMAX1(T1,T3)*CT, AND
C                IF IVAR IS SET AT 0, THEN IVAR = AMAX1(FCF,SCF)*2.*PI*
C                  20.*TMAX+1.
C    FOR COMPOSITE FILTERS, PARAMETERS FOR THE SECOND FILTER ARE INPUT
C    THROUGH NAMELIST /FIL2/ AND ARE AS FOLLOWS:
C          KF       NP        FCF
C          BW       RF        SEE NAMELIST /VALUE/ FOR EXPLANATION FO
C    PARAMETERS.
C    FORMAT OF NAMELIST INPUT IS:
C         COL 1        BLANK
C         COL 2-7  '$VALUE' OR '$FIL2 '
C         COL 8      BLANK
C         COL 9-80 PARAMETER NAME FOLLOWED BY EQUAL SIGN FOLLOWED BY
C                  PARAMETER VALUE.  SEPERATE PARAMETERS BY COMMAS AND
C                  END WITH A DOLLAR SIGN.
C    IF ONE CARD IS INSUFFICIENT - INPUT MAY BE CONTINUED ON OTHER
C    CARDS (START IN COL 2), BREAKING PARAMETER LIST AT ANY COMMA.
C--------------------------------------------------------------------
      CALL PLOTS (DATA,1024,4)
      CALL PLOT(0.,5.,-3)
      WRITE (6,2000)
1     TSTART=SECOND(CP)
      READ(5,VALUE)
      FF(1)=KF   $FF(2)=BW   $FF(3)=NP   $FF(4)=RF   $FF(5)=FCF
5     CONTINUE
      IF(NP.GT.12) GO TO 700
      WRITE (6,VALUE)
      OMI=2.*PI*FCF
      OMEGAI=CMPLX(0.,OMI)
```

```
      OMJ=2.*PI*SCF
      OMEGAJ=CMPLX(0.,OMJ)
      NP2=NP*2
      TERM1=BW*2.*PI
      M=0
      GO TO (10,20),KF
C-----------------------------------------------------------------------
C   CALCULATION OF "P" VALUES FOR BUTTERWORTH FILTER
C-----------------------------------------------------------------------
   10 DO 11 I=1,NP
      TERM2=FLOAT(2*I-1+NP)/FLOAT(NP2)*PI
      TPR=-TERM1*COS(TERM2)
      TPI=TERM1*SIN(TERM2)
      P(I)=CMPLX(TPR,TPI)
   11 WRITE (6,3000) I,P(I)
      GO TO 30
C-----------------------------------------------------------------------
C   CALCULATION OF "P" VALUES FOR CHEBYSHEV FILTER
C-----------------------------------------------------------------------
   20 EI=1./SQRT(10.**(RF/10.)-1.)
      V=1./FLOAT(NP)*SINHI(EI)
      DO 21 I=1,NP
      U=PI/FLOAT(NP2)*FLOAT(2*I-1)
      TPR=TANH(V)*SIN(U)*TERM1
      TPI=COS(U)*TERM1
      P(I)=CMPLX(TPR,TPI)
   21 WRITE (6,3000) I,P(I)
C-----------------------------------------------------------------------
C   CALCULATION OF "S" VALUES
C-----------------------------------------------------------------------
   30 J=0
      DO 31 I=2,NP2,2
      J=J+1
      IM1=I-1
      S(I-1)=P(J)/2.+OMEGAI
      S(I)=P(J)/2.-OMEGAI
   31 WRITE (6,3001) IM1,S(I-1),I,S(I)
      GO TO (35,400,450)ICF
C-----------------------------------------------------------------------
C   CALCULATION OF "A" VALUES
C-----------------------------------------------------------------------
35    CONTINUE
      DO 32 I=1,NP2
      ST=-S(I)
      TERM4=CMPLX(1.,0.)
      DO 33 J=1,NP2
      IF (I.EQ.J) GO TO 33
      TERM4=TERM4*(ST+S(J))
   33 CONTINUE
      ST2=ST**NP
```

```
C     WRITE(6,2001)"ST2=",ST2,"TERM4=",TERM4
2001  FORMAT(1X,A4,2E19.10,5X,A6,2E19.10)
      A(I)=ST2/TERM4
   32 WRITE (6,3002) I,A(I)
C-----------------------------------------------------------------------
C   CALCULATION OF "IVAR" AND "TMAX" IF NECESSARY
C-----------------------------------------------------------------------
      IF (FLOAT(IVAR).GT.0..AND.TMAX.GT.0.) GO TO 40
      IF (TMAX.GT.0.) GO TO 41
      TMAX=AMAX1(T1,T3)*CT
      IF (IVAR.GT.0.) GO TO 40
   41 IVAR=AMAX1(OMI,OMJ)*5.*TMAX+1.
   40 DELT=TMAX/FLOAT(IVAR-1)
      WRITE (6,2002) IVAR,DELT
      DELT2=.8*AMIN1(1./FCF,1./SCF)
      T=DELT
      Y(1)=CMPLX(0.,0.)
      GO TO (100,200,300),KIS
C-----------------------------------------------------------------------
C   COMPOSITE FILTERS
C-----------------------------------------------------------------------
400   DO 410 I=1,NP2
410   SS(I)=S(I)
      ICF=ICF+1
      READ(5,FIL2)
      SF(1)=KF     $SF(2)=BW     $SF(3)=NP   $SF(4)=RF     $SF(5)=FCF
      GO TO 5
450   J=NP2+2*FF(3)
      IF(J.GT.48)GO TO 700
      K=NP2+1       $NSP=0
      DO 460 I=K,J
      NSP=NSP+1
460   S(I)=SS(NSP)
      IF(FF(2).LT.BW)BW=FF(2)
      NP=J/2
      NP2=J
      WRITE(6,3001)(J,S(J),J=1,NP2)
      GO TO 35
C-----------------------------------------------------------------------
C   CALCULATION OF OUTPUT SIGNAL FOR SQUARE WAVE INPUT SIGNAL
C-----------------------------------------------------------------------
  100 DO 101 I=2,IVAR
      L=MIN0(3,I)
      Y(L)=CMPLX(0.,0.)
      TT=SNGL(T)
      DO 102 J=1,NP2
      X1=CEXP(-S(J)*TT)
C     PRINT *,"X1=",X1,"     T=",TT
      X2=S(J)+OMEGAJ
      X3=S(J)-OMEGAJ
      IF(TT.GT.T1)GO TO 103
```

```
      X4=OMEGAJ*TT
C     PRINT *," X4=",X4
      Y(L)=Y(L)+A(J)/2.*((CEXP(X4)-X1)/X2+(CEXP(-X4)-X1)/X3)
      GO TO 102
103   Y(L)=Y(L)+A(J)/2.*X1*((CEXP(X2*T1)-1.)/X2+(CEXP(X3*T1)-1.)/X3)
  102 CONTINUE
      IF(.NOT.IY) GO TO 105
      IM1=I-1
      IF (I.EQ.2) WRITE (6,2003) IM1,Y(IM1)
      WRITE (6,2015) I,Y(L)
C-------------------------------------------------------------------------
C  CALCULATION OF ENEVELOP
C-------------------------------------------------------------------------
105   CALL ARRAY
      IF(T.GT.TMAX)GO TO 500
  101 T=T+DELT
      GO TO 500
C-------------------------------------------------------------------------
C  CALCULATION OF OUTPUT SIGNAL FOR COMPOSITE WAVE INPUT SIGNAL
C-------------------------------------------------------------------------
  200 D=1./T1
      E=1./(T3-T2)
      DO 211 I=1,NP2
      B(I)=S(I)+OMEGAJ
      C(I)=S(I)-OMEGAJ
  211 WRITE (6,3004) I,B(I),I,C(I)
      WRITE (6,3003) D,E
      DO 201 I=2,IVAR
      TT=SNGL(T)
      L=MIN0(3,I)
      Y(L)=CMPLX(0.,0.)
      DO 202 J=1,NP2
      X1=CEXP(-S(J)*TT)
      X2=CEXP(OMEGAJ*TT)
      X3=CEXP(-OMEGAJ*TT)
      X5=1./C(J)**2
      X4=1./B(J)**2
      IF(TT.LE.T1)GO TO 203
      X6=CEXP(B(J)*TT)
      X7=CEXP(C(J)*TT)
      X61=CEXP(B(J)*T1)
      X71=CEXP(C(J)*T1)
      IF(TT.LE.T2)GO TO 204
      X62=CEXP(B(J)*T2)
      X72=CEXP(C(J)*T2)
      IF(TT.LE.T3)GO TO 205
      X63=CEXP(B(J)*T3)
      X73=CEXP(C(J)*T3)
      GO TO 206
203   Y(L)=Y(L)+D*A(J)/2.*((X1-X2)*X4+X2*TT/B(J)+(X1-X3)*X5+X3*TT/C(J))
      GO TO 202
```

```
204   Y(L)=Y(L)+D*A(J)/2.*X1*((X61*(T1/B(J)-X4)+X71*(T1/C(J)-X5))+
    1  X4+X5+T1*((X6-X61)/B(J)+(X7-X71)/C(J)))
      GO TO 202
205   Y(L)=Y(L)+A(J)/2.*X1*(D*(X4+X5+X61*(T1/B(J)-X4)+X71*(T1/C(J)-X5)
    1  +T1*((X6-X61)/B(J)+(X7-X71)/C(J)))-E*(X6*(T/B(J)-X4)+X7*(T/
    2  C(J)-X5)-X62*(T2/B(J)-X4)-X72*(T2/C(J)-X5)-T2*((X6-X62)/
    3  B(J)+(X7-X72)/C(J))))
      GO TO 202
206   Y(L)=Y(L)+A(J)/2.*X1*(D*(X4+X5+X61*(T1/B(J)-X4)+X71*(T1/C(J)-X5)
    1  +T1*((X6-X61)/B(J)+(X7-X71)/C(J)))+E*(X62*(T2/B(J)-X4)+X72*
    2  (T2/C(J)-X5)+T2*((X6-X62)/B(J)+(X7-X72)/C(J))-X63*(T3/B(J)-
    3  X4)-X73*(T3/C(J)-X5)-T3*((X6-X63)/B(J)+(X7-X73)/C(J))))
  202 CONTINUE
      IF(.NOT.IY) GO TO 215
      IM1=I-1
      IF (I.EQ.2) WRITE (6,2003) IM1,Y(IM1)
      WRITE (6,2015) I,Y(L)
C-----------------------------------------------------------------------
C  CALCULATION OF ENEVELOP
C-----------------------------------------------------------------------
215   CALL ARRAY
      IF(T.GT.TMAX)GO TO 500
  201 T=T+DELT
      GO TO 500
C-----------------------------------------------------------------------
C  CALCULATION OF OUTPUT SIGNAL FOR SINE SQUARED WAVE INPUT SIGNAL
C-----------------------------------------------------------------------
  300 DO 305 I=1,NP2
      B(I)=S(I)+OMEGAJ
      C(I)=S(I)-OMEGAJ
  305 WRITE (6,3004) I,B(I),I,C(I)
      DO 301 I=2,IVAR
      L=MIN0(3,I)
      Y(L)=CMPLX(0.,0.)
      TT=SNGL(T)
      DO 302 J=1,NP2
      IF(TT.GE.T1)GO TO 303
      X1=CEXP(-OMEGAJ*TT)
      X2=CEXP(OMEGAJ*TT)
      R1=T1/PI
      X3=B(J)*R1
      X4=C(J)*R1
      D1=PI*TT/T1
      D2=SIN(D1)
      D3=COS(D1)
      D4=X3*D2*D2-2.*D2*D3
      X8=CEXP(-S(J)*TT)
      X5=2.*PI/(B(J)*T1)
      X6=X2/(2.*X3*X3+8.)*(D4+X5)
      D4=X4*D2*D2-2.*D2*D3
      X7=X1/(2.*X4*X4+8.)*(D4+2.*PI/(C(J)*T1))
```

```
      X88=X8*PI**3     $X9=T1**3
      R2=4.*PI*PI*T1
      Y(L)=Y(L)+A(J)*(X6+X7-X88/(B(J)**3*X9+R2*B(J))-X88/(X9*C(J)**3
     1 +R2*C(J)))
      GO TO 302
303   X1=A(J)*CEXP(-S(J)*TT)*PI
      X2=B(J)*T1
      X3=B(J)**3
      R1=T1**3
      R2=PI*PI
      R3=R1/R2
      X4=CEXP(X2)/(X3*R3+4.*X2)
      X5=R2/(X3*R1+4.*R2*X2)
      X2=C(J)*T1
      X3=C(J)**3
      X6=CEXP(X2)/(X3*R3+4.*X2)
      X7=R2/(X3*R1+4.*R2*X2)
      Y(L)=Y(L)+X1*(X4+X6-X5-X7)
  302 CONTINUE
      IF(.NOT.IY) GO TO 306
      IM1=I-1
      IF (I.EQ.2) WRITE (6,2003) IM1,Y(IM1)
      WRITE (6,2015) I,Y(L)
C-----------------------------------------------------------------------
C  CALCULATION OF ENEVELOP
C-----------------------------------------------------------------------
306   CALL ARRAY
      IF(T.GT.TMAX)GO TO 500
  301 T=T+DELT
500   IF(.NOT.IE) GO TO 600
      WRITE(6,2004)
      DO 503 I=1,M
  503 WRITE (6,2005) I,AH(I),I,TH(I)
C-----------------------------------------------------------------------
C  DETERMINATION OF ABSOLUTE MAXIMUM AMPLITUDE OF ARRAY
C-----------------------------------------------------------------------
600   CALL MAXAMP
      WRITE (6,2008) AMPT
C-----------------------------------------------------------------------
C  CALCULATION OF ARRAY COORDINATES FOR CALCOMP PLOTTER
C-----------------------------------------------------------------------
      CALL COORD
C-----------------------------------------------------------------------
C  OFFLINE CALCOMP PLOTTING ROUTINE
C-----------------------------------------------------------------------
602   CALL GRAPH(KIS,SCF,T1,T2,T3,ICF,FF,SF)
      WRITE (6,VALUE)
      IF(IEND) GO TO 4
    2 WRITE (6,2013)
      CALL PLOTE(NAME)
      WRITE (6,2014) NAME
```

```
      TSTOP=SECOND(CP)
      DELTAT=TSTOP-TSTART
      WRITE (6,4000) TSTART,TSTOP,DELTAT
      STOP
    4 IVAR=0
      TMAX=0.
      ICF=1
      TSTOP=SECOND(CP)
      DELTAT=TSTOP-TSTART
      WRITE (6,4000) TSTART,TSTOP,DELTAT
      GO TO 1
700   WRITE(6,4001)
      STOP
4001  FORMAT(*0THE NUMBER OF POLES OF THE FILTER EXCEEDS THE ALLOCATED
     1DIMENSION*/)
 2000 FORMAT (1H1,*THE FOLLOWING PORTION OF THIS PRINT OUT IS THE OUTPUT
     A SECTION*)
 2002 FORMAT (1H0////* IVAR = *,I7,10X,* DELT = *,1PD17.10/)
 2003 FORMAT (1H1,*THE FOLLOWING OUTPUT IS THE COMPLEX Y FOR EACH DELTA
     ATIME INCREMENT*///* Y(*,I6,*) = *,1P2E19.10)
 2004 FORMAT (1H1,*THE FOLLOWING OUTPUT IS THE ARRAY OF LOCAL MAXIMUMS (
     ASUBSET OF COMPLEX VARIABLES) AND TIME*/)
 2005 FORMAT (1H ,*AH(*,I4,*) = *,1PE19.10,*        TH(*,I4,*) = *,E19.10)
 2006 FORMAT (1H1,*THE FOLLOWING OUTPUT IS THE ARRAY OF LOCAL MINIMUMS (
     ASUBSET OF COMPLEX VARIABLES) AND TIME*/)
 2007 FORMAT (1H ,*AL(*,I4,*) = *,1PE19.10,*        TL(*,I4,*) = *,E19.10)
 2008 FORMAT (1H0///* ABSOLUTE MAXIMUM AMPLITUDE IS EQUAL TO *,1PE17.10)
 2013 FORMAT (1H0////* THIS STATEMENT TERMINATES THIS RUN*)
 2014 FORMAT (1H0///* THE NUMBER OF BLOCKS ON THE PLOTTING TAPE IS *,I5)
 2015 FORMAT (1H ,*Y(*,I6,*) = *,1P2E19.10)
 3000 FORMAT (1H0,*P(*,I2,*) =*,1P2E19.10)
 3001 FORMAT (1H0,*S(*,I2,*) =*,1P2E19.10,*       S(*,I2,*) =*,2E19.10)
 3002 FORMAT (1H0,*A(*,I2,*) =*,1P2E19.10)
 3003 FORMAT (1H0,*D = *,1PE19.10,*        E =*,E19.10)
 3004 FORMAT (1H0,*B(*,I2,*) =*,1P2E19.10,*       C(*,I2,*) =*,2E19.10)
 4000 FORMAT (1H0,*THIS RUN STARTED AT TIME *,F7.3,* AND ENDED AT *,F7.3
     A,* TAKING *,F7.3,* SECONDS TO EXECUTE*///)
      END
      SUBROUTINE ARRAY
C-----------------------------------------------------------------------
C  ROUTINE TO LIMIT OUTPUT OF ENVELOP ARRAYS
C-----------------------------------------------------------------------
      COMPLEX Y(3)
      DOUBLE DELT,T
      DIMENSION AH(2000),TH(2000)
      COMMON/VARS/AH,AMPT,DELT,IVAR,L,M,T,TH,TMAX,XVAR,XVAR1,
     AYVAR,Y,DELT2,IFP,IFA
      IF (L.EQ.2) GO TO 2
      IF (L.NE.3) GO TO 7
      YMAX=AMAX1(REAL(Y(L-2)),REAL(Y(L-1)),REAL(Y(L)))
```

```
      IF (YMAX.EQ.REAL(Y(L-1))) GO TO 3
      GO TO 5
3     IF(Y(L-2).EQ.CMPLX(0.,0.))GO TO 5
      M=M+1
      IF(M.GT.2000) GO TO 6
      TH(M)=SNGL(T)
      AH(M)=REAL(Y(L-1))
      IF(M.LE.30)GO TO 5
      Y(L-1)=Y(L)=CMPLX(0.,0.)
      T=T+DELT2-DELT
C     WRITE(6,3000)Y(L-1),TH(M)
      GO TO 5
2     M=1
      TH(M)=SNGL(T-DELT)
      AH(M)=REAL(Y(L-1))
      RETURN
    5 Y(L-2)=Y(L-1)
      Y(L-1)=Y(L)
      RETURN
    6 WRITE (6,2000)
      STOP
    7 WRITE (6,2001)
      STOP
 2000 FORMAT (1H1,*THE PLOTTING ARRAY EXCEEDS THE ALLOCATED DIMENSION OF
     A 2000*)
 2001 FORMAT (1H1,*ERROR IN THE CALCULATION OF L COUNTER*)
3000  FORMAT(5X,*Y=*,1P2E19.10,*J   MAX   T=*,E19.10)
3001  FORMAT(5X,*Y=*,1P2E19.10,*J   MIN   T=*,E19.10)
      END
      SUBROUTINE MAXAMP
C-----------------------------------------------------------------------
C  ROUTINE TO CALCULATE ABSOLUTE MAX SIGNAL AMPLITUDE
C-----------------------------------------------------------------------
      COMPLEX Y(3)
      DOUBLE DELT,T
      DIMENSION AH(2000),TH(2000)
      COMMON/VARS/AH,AMPT,DELT,IVAR,L,M,T,TH,TMAX,XVAR,XVAR1,
     AYVAR,Y,DELT2,IFP,IFA
      AMPT=AH(1)
      DO 1 I=2,M
    1 AMPT=AMAX1(AMPT,AH(I))
      RETURN
      END
      SUBROUTINE COORD
C-----------------------------------------------------------------------
C  ROUTINE TO CALCULATE PLOTTER COORDINATES IN INCHES
C-----------------------------------------------------------------------
      COMPLEX Y(3)
      DOUBLE DELT,T
      DIMENSION AH(2000),TH(2000)
      COMMON/VARS/AH,AMPT,DELT,IVAR,L,M,T,TH,TMAX,XVAR,XVAR1,
     AYVAR,Y,DELT2,IFP,IFA
```

```
      XVAR1=XVAR
2     IF(FLOAT(M-1).LE.XVAR1*100.)GO TO 1
      XVAR1=XVAR1*1.1
      GO TO 2
    1 WRITE (6,2000) XVAR1,YVAR
 2000 FORMAT (1H0////* LENGTH OF THE X-AXIS IS *,F5.3,* INCHES, AND THE
     AHEIGHT OF THE Y-AXIS IS *,F6.3,* INCHES*)
      END
      SUBROUTINE GRAPH(KIS,SCF,T1,T2,T3,ICF,FF,SF)
C-----------------------------------------------------------------------
C  ROUTINE TO PLOT ARRAYS
C-----------------------------------------------------------------------
      COMPLEX Y(3)
      DOUBLE DELT,T
      LOGICAL IFP,IFA
      DIMENSION FF(5),SF(5)
      DIMENSION AH(2000),TH(2000)
      DIMENSION IT(8),ITL(5)
      COMMON/VARS/AH,AMPT,DELT,IVAR,L,M,T,TH,TMAX,XVAR,XVAR1,
     1YVAR,Y,DELT2,IFP,IFA
      DATA(IT(I),I=1,8)/6HSQUARE,9HCOMPOSITE,10HSINE SQUAR,2HED,
     1  10HBUTTERWORT,1HH,9HCHEBYSHEV,5HPOLES/
      DATA(ITL(I),I=1,5)/6,9,12,11,9/
      AH(M+1)=TH(M+1)=0.
      TH(M+2)=TMAX/XVAR1
      AH(M+2)=AMPT/2.5
      CALL AXIS(0.,-YVAR,15HTIME (MSECONDS),-15,XVAR1,0.,TH(M+1),TH(M+2)
     1)
      TYV=YVAR*2.
      CALL AXIS (0.,-YVAR,18HRELATIVE AMPLITUDE,18,TYV,90.,-1.,.4)
      CALL PLOT (0.,0.,3)
      CALL PLOT (XVAR1,0.,2)
      CALL PLOT (0.,0.,3)
      CALL LINE(TH,AH,M,1,0.,4)
      DO 2 I=1,M
2     AH(I)=-AH(I)
      CALL LINE(TH,AH,M,1,0.,4)
      XPLUS=XVAR1+3.25
C  INPUT SIGNAL
      H=.14
      CALL SYMBOL(0.5,-3.7,H,13HINPUT SIGNAL:,0.,13)
      CALL SYMBOL(2.37,-3.7,H,IT(KIS),0.,ITL(KIS))
      CALL SYMBOL(2.37,-3.9,H,3HF =,0.,3)
      CALL SYMBOL(2.44,-3.9,.07,1HC,0.,1)
      FC=SCF
      CALL NUMBER(2.77,-3.9,H,FC,0.,2)
      CALL SYMBOL(3.57,-3.9,H,3HKHZ,0.,3)
      CALL SYMBOL(2.37,-4.1,H,3HT1=,0.,3)
      XT=T1
      CALL NUMBER(2.77,-4.1,H,XT,0.,3)
      CALL SYMBOL(3.57,-4.1,H,4HMSEC,0.,4)
```

```
      IF(T2.EQ.0.)GO TO 5
      CALL SYMBOL(2.37,-4.3,H,3HT2=,0.,3)
      XT=T2
      CALL NUMBER(2.77,-4.3,H,XT,0.,3)
      CALL SYMBOL(3.57,-4.3,H,4HMSEC,0.,4)
      IF(T3.EQ.0.)GO TO 5
      CALL SYMBOL(2.37,-4.5,H,3HT3=,0.,3)
      XT=T3
      CALL NUMBER(2.77,-4.5,H,XT,0.,3)
      CALL SYMBOL(3.57,-4.5,H,4HMSEC,0.,4)
C     FILTER
5     CALL SYMBOL(4.5,-3.7,H,7HFILTER:,0.,7)
      KF=FF(1)
      CALL SYMBOL(5.6,-3.7,H,IT(2*KF+3),0.,ITL(KF+3))
      CALL SYMBOL(5.6,-3.9,H,3HF =,0.,3)
      CALL SYMBOL(5.73,-3.9,.07,1HC,0.,1)
      CALL NUMBER(6.1,-3.9,H,FF(5),0.,2)
      CALL SYMBOL(7.1,-3.9,H,3HMHZ,0.,3)
      CALL NUMBER(5.6,-4.1,H,FF(3),0.,-1)
      CALL SYMBOL(5.9,-4.1,H,IT(8),0.,5)
      CALL SYMBOL(5.6,-4.3,H,3HBW=,0.,3)
      CALL NUMBER(6.1,-4.3,H,FF(2),0.,2)
      CALL SYMBOL(7.1,-4.3,H,3HMHZ,0.,3)
      IF(FF(4).EQ.0.)GO TO 8
      CALL SYMBOL(5.6,-4.5,H,5HRF = ,0.,5)
      CALL NUMBER(6.2,-4.5,H,FF(4),0.,2)
8     IF(ICF.EQ.1)GO TO 10
      X=7.5     $XY=-3.7
      CALL SYMBOL(X,XY,H,7HFILTER:,0.,7)
      X=X+1.1      $KF=SF(1)
      CALL SYMBOL(X,XY,H,IT(2*KF+3),0.,ITL(KF+3))
      XY=XY-.2
      CALL SYMBOL(X,XY,H,3HF =,0.,3)
      CALL SYMBOL(X+.13,XY,.07,1HC,0.,1)
      FC=SF(5)
      CALL NUMBER(X+.5,XY,H,FC,0.,2)
      CALL SYMBOL(X+1.5,XY,H,3HMHZ,0.,3)
      XY=XY-.2
      CALL NUMBER(X,XY,H,SF(3),0.,-1)
      CALL SYMBOL(X+.3,XY,H,IT(8),0.,5)
      XY=XY-.2
      CALL SYMBOL(X,XY,H,3HBW=,0.,3)
      FC=SF(2)
      CALL NUMBER(X+.5,XY,H,FC,0.,2)
      CALL SYMBOL(X+1.5,XY,H,3HMHZ,0.,3)
      IF(SF(4).EQ.0.)GO TO 10
      XY=XY-.2
      CALL SYMBOL(X,XY,H,3HRF=,0.,3)
      CALL NUMBER(X+.6,XY,H,SF(4),0.,2)
10    IF(.NOT.IFP)GO TO 40
      IM=M-1
```

```
      DO 20 I=1,IM
20    AH(I)=1./(TH(I+1)-TH(I))-SCF
      AH(M)=AH(M+1)=AH(10)
      IM=M-1
      DO 30 I=10,IM
      AH(M)=AMIN1(AH(M),AH(I))
30    AH(M+1)=AMAX1(AH(M+1),AH(I))
      AH(M+1)=(AH(M+1)-AH(M))/2.5
      PRINT *,"  AH(M)=",AH(M),"        AH(M+1)=",AH(M+1)
      TH(M)=0.
      TH(M+1)=TMAX/XVAR1
      CALL AXIS(XVAR1,0.,15HFREQUENCY IN HZ,-15,YVAR,90.,AH(M),AH(M+1))
      CALL DASHLN(TH(10),AH(10),IM-9,1)
      IF(.NOT.IFA)GO TO 40
      WRITE(6,50)(I,AH(I),I,TH(I),I=10,IM)
50    FORMAT(*1 THE FOLLOWING OUTPUT IS COMPUTED SIGNAL FREQUENCY*
     1 (*  FP(*,I4,*) = *,E19.10,*   TH(*,I4,*) = *,E19.10))
40    CALL PLOT(XPLUS,0.,-3)
      RETURN
      END
**
```


☆U.S.Government Printing Office: 1978 — 757-080/569